APPENDIX C

COMPUTER PROGRAMS FOR DETERMINING COMPUTATIONAL ERRORS, OUTPUT RIPPLE, AND 1% SETTLING TIME OF RMS CONVERTERS

Introduction

Rather than use filter charts or formulas, many people may prefer to use a computer program to perform calculations. The following programs have been written for the Apple II computer. The first calculates values of dc error, ripple, and averaging error for each of the three basic connections (i.e., C_{AV} only, one pole post filter, two pole Sallen-Key filter).

For the one pole filter case, output values using C_2 equal to 2.2 times C_{AV} and 3.3 times C_{AV} will be automatically printed out when the one pole filter case is selected. For the two pole filter, capacitors C_2 and C_3 are both set equal to 2.2 times C_{AV} (see filters and averaging section – page 12).

The second computer program calculates the total settling time for each connection using the values of C_{AV}, and/or C_2 and C_3 selected.

PROGRAM #1

RMS CONVERTER RIPPLE/ERROR PROGRAM

```
10  REM FROM ANALOG DEVICES SEMICONDUCTOR
20  REM WRITTEN FOR APPLE II WITH PRINTER CARD IN SLOT #1
30  REM NOV. 6, 1985
40  REM PROGRAM TO COMPUTE %D.C. ERROR AND % RIPPLE IN RMS CONVERTERS
50  DIM E(20),F(20),R(20),R2(20),R3(20),AV(20),A2(20)
60  HOME : VTAB 4: PRINT "RMS CONVERTER ERROR/RIPPLE PROGRAM"
70  PRINT "**********************************": PRINT
80  PRINT "IS PRINTER AVAILABLE ? ";: GOSUB 560:X2 = X:X3 = X
82  PRINT : PRINT "VALUES INDICATED FOR C2 AND C3 ARE FOR": PRINT "THE AD6
     37 AND AD536A RMS CONVERTERS.": PRINT
84  PRINT "FOR THE AD636, VALUES FOR THESE": PRINT "CAPACITORS MUST BE 2.5
      TIMES THE AMOUNT": PRINT "SHOWN FOR SAME % RIPPLE AND AVG ERROR.": PRINT

90  INPUT " ENTER CAVG VALUE IN UF : ";C1
100 M = .025:T = (M * C1):M1 = .0245:PL = 0:D$ =  CHR$ (4)
110 C(2) = 2.2 * C1:C(3) = 3.3 * C1:C4 = 2
120  REM M & M1 = 25K & 24.5K OHMS, ADJUSTED SO CAP VALUES ARE IN UF; PL=
     # OF POLES.
130  PRINT "ARE YOU USING AN OUTPUT FILTER ? ";: GOSUB 560
140  IF X = 1 THEN  INPUT " IS THIS A 1 OR 2 POLE FILTER ? ";PL
150  READ SAMPLES: FOR S = 1 TO SAMPLES: READ F(S)
160 E(S) = 1 / (0.16 + (6.4 * (T ↑ 2) * (F(S) ↑ 2))):RD = E(S): GOSUB 550:
     E(S) = RD
170 R(S) = 50 / ((1 + ((40) * (T ↑ 2) * (F(S) ↑ 2))) ↑ 0.5):RD = R(S): GOSUB
     550:R(S) = RD
180  ON PL + 1 GOTO 190,210,250
190  REM CAVG ONLY CALCULATIONS
```

```
200 AV(S) = R(S) + E(S): GOTO 290
210  REM 1 POLE CALCULATIONS
220 R2(S) = R(S) / (1 + (12.57 * F(S) * C(2) * M) ↑ 2) ↑ .5:RD = R2(S): GOSUB
     550:R2(S) = RD
230 R3(S) = R(S) / (1 + (12.57 * F(S) * C(3) * M) ↑ 2) ↑ .5:RD = R3(S): GOSUB
     550:R3(S) = RD
240 AV(S) = R2(S) + E(S):A2(S) = R3(S) + E(S): GOTO 290
250  REM 2 POLE SALLEN-KEY CALCULATIONS
260 R3(S) = R(S) / (M1 * C(2)) ↑ 2 / ((1 / (M1 * C(2)) ↑ 2 - (12.57 * F(S)
     ) ↑ 2) ↑ 2 + (25.13 * F(S) / (M1 * C(2))) ↑ 2) ↑ .5
270 RD = R3(S): GOSUB 550:R3(S) = RD
280 AV(S) = R3(S) + E(S)
290  NEXT S
300 X = 2: GOTO 320: REM PRINTOUT RESULTS
310  PRINT D$;"PR# 1": PRINT  CHR$ (29): REM SELECT PRINTER & AMOUNT OF CH
     AR./INCH (VARIES WITH PRINTER)
320  HOME : PRINT "OUTPUT RIPPLE AND AVERAGING ERROR ARE": PRINT "CALCULAT
     ED FOR ";
330  ON PL + 1 GOTO 340,350,360
340  PRINT C1;" UF CAVG ONLY": GOTO 370
350  PRINT "C2= ";C(C4);"UF, CAV= ";C1;"UF": GOTO 370
360  PRINT "2 POLE SALLEN-KEY FILTER": PRINT "C2 AND C3 =";C(2);" UF, CAVG
     = ";C1;" UF"
370  PRINT : PRINT "FREQUENCY DC ERROR  RIPPLE    AVG ERROR"
380  FOR S = 1 TO SAMPLES
390 B$ =  STR$ (F(S)) + " HZ":B = 10 -  LEN (B$):B2$ =  STR$ (E(S)) + "%":
     B2 = 10 -  LEN (B2$)
400  ON PL + 1 GOTO 410,420,440
410 B3$ =  STR$ (R(S)) + "%":B4$ =  STR$ (AV(S)) + "%": GOTO 450
420  IF C4 = 2 THEN B3$ =  STR$ (R2(S)) + "%":B4$ =  STR$ (AV(S)) + "%": GOTO
     450
430  IF C4 = 3 THEN B3$ =  STR$ (R3(S)) + "%":B4$ =  STR$ (A2(S)) + "%": GOTO
     450
440 B3$ =  STR$ (R3(S)) + "%":B4$ =  STR$ (AV(S)) + "%"
450 B3 = 10 -  LEN (B3$): PRINT B$; SPC( B);B2$; SPC( B2);B3$; SPC( B3);B4
     $
460  NEXT
470  IF PL < > 1 OR C4 < > 2 THEN 500
480  PRINT : PRINT : IF X = 1 THEN C4 = C4 + 1: GOTO 320
490  PRINT D$;"PR# 0": PRINT "HIT ANY KEY TO CONTINUE";: GET A$:C4 = C4 +
     1: GOTO 320
500  ON X2 GOTO 510,520
510  PRINT D$;"PR# 0": PRINT : PRINT "DO YOU WANT A PRINTOUT ? ";: GOSUB 5
     60:C4 = 2:X2 = 2: IF X = 1 THEN 310
520  PRINT : PRINT D$;"PR# 0": PRINT "TRY OTHER VALUES ? ";:X2 = X3: GOSUB
     560
530  IF X = 2 THEN  END
540  RESTORE : GOTO 90
550 RD =  INT ((RD * 1000) + .5) / 1000: RETURN : REM ROUND ROUTINE
560  GET E$:X = 2: IF E$ = "Y" THEN  PRINT "YES":X = 1: RETURN
570  IF E$ < > "N" THEN X = 3: PRINT : PRINT "PLEASE REENTER : ";: GOTO 5
     60
580  PRINT "NO": RETURN
590  DATA  11,1,2,4,10,20,40,60,100,200,400,1000
600  REM 1ST PIECE OF DATA CONTAINS AMOUNT OF SAMPLES TO BE READ (20 MAX)
610  REM REMAINDER OF DATA IS FREQUENCY IN HZ
```

PROGRAM #2

RMS CONVERTER COMBINED SETTLING TIME PROGRAM

```
10   REM  FROM ANALOG DEVICES SEMICONDUCTOR
20   REM  WRITTEN BY CHARLES KITCHIN AND ANDREW WHEELER MARCH 22,'83
30   REM PROGRAM TO CALCULATE 1% SETTLE TIME FOR SINEWAVE BURST
40   PRINT "*****************************"
50   PRINT  SPC( 11);"AD637"
60   HOME : PRINT "COMBINED SETTLING TIME PROGRAM"
70   PRINT "*****************************"
80   INPUT "ENTER CAVG (C1) VALUE IN UF : ";CAV
90   INPUT "ENTER C2 VALUE (NO C2= 0UF) : ";C2
100   INPUT "ENTER VALUE OF C3 (NO C3= 0UF) : ";C3
110 S(1) = .025 * CAV * 4.6
120 S(2) =  SQR ((S(1) ^ 2) + ((.025 * C2 * 4.6) ^ 2))
130 S(3) =  SQR ((S(1) ^ 2) + ((.025 * C2 * 4.6) ^ 2) + ((.024 * C3 * 4.6
    ) ^ 2))
140   FOR A = 1 TO 3:S(A) =  INT ((S(A) * 10000) + .5) / 10000: NEXT
150   PRINT "THE SETTLING TIMES EQUAL...": PRINT
160   PRINT S(1);" SECONDS FOR CAV ONLY"
170   IF C2 = 0 THEN 210
180   PRINT : PRINT S(2);" SECONDS FOR CAV & C2"
190   PRINT : IF C3 = 0 THEN 210
200   PRINT S(3);" SECONDS FOR CAV AND": PRINT " 2 POLE SALLEN-KEY FILTER.
      "
204   PRINT : PRINT "NOTE:"
205   PRINT "THESE VALUES ARE FOR THE AD637 AT ALL INPUT SIGNAL LEVELS"
206   PRINT "HOWEVER,FOR THE AD536A AND AD636,SETTLING TIME INCREASES AT L
      OW LEVELS.THIS INCREASE BEGINS AT INPUT LEVELS BELOW 200MV FOR THE A
      D536A AND AT LEVELS BELOW 80 MV WHEN USING THE AD636"
210   PRINT : PRINT "PERFORM ANOTHER CALCULATION ? ";: GET A$
220   IF A$ = "Y" THEN  PRINT "YES": GOTO 80
230   IF A$ <  > "N" THEN  PRINT : GOTO 210
240   PRINT "NO": END
```